
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of several test cases, which are part of the LTE/SAE test suite. 

Corrections required for LTE/SAE iwd_10wk26 ATS
3.1 Change 1  /  f_EUTRA_FillNAS_SecurityCapability corrected for UTRAN + GERAN security algorithms
	Function name
	f_EUTRA_FillNAS_SecurityCapability

	Reason for change
	The replying of UTRAN and GERAN security algorithm capabilities in the security mode command is not correct.

TTCN does not take into account when a UE sends GEA but no UxA in Attach Request. Additionally, it does not respect the following statement from 24.301 clause 9.9.3.36: “If octet 5 is included, then also octet 6 shall be included.”
 

	Summary of change
	Corrected to set octet 6 to ‘00’O in case octet 5 is set.


Corrected to set octets 5 and 6 to ‘00’O in case that only GERAN algorithm capability is reported and hence octet 7 needs to be set.


	TTCN module
	EUTRA_NASSteps.ttcn

	MCC-STF160
	


Before:

	  function f_EUTRA_FillNAS_SecurityCapability (ATTACH_REQUEST p_ATTACH_REQUEST) return UE_SecurityCapability

  {

    var UE_SecurityCapability v_SecurityCapability;

    var B8_Type v_GEA := '00000000'B;

    var integer v_Length := 2;

    v_SecurityCapability.eeaCap := p_ATTACH_REQUEST.ueNetworkCapability.eeaCap;

    v_SecurityCapability.eiaCap := p_ATTACH_REQUEST.ueNetworkCapability.eiaCap;

    if (ispresent(p_ATTACH_REQUEST.ueNetworkCapability.ueaCap)) {

      v_SecurityCapability.ueaCap := p_ATTACH_REQUEST.ueNetworkCapability.ueaCap;

      v_Length := v_Length + 1;
      if (ispresent(p_ATTACH_REQUEST.ueNetworkCapability.uiaCap)) {

        v_SecurityCapability.uiaCap := p_ATTACH_REQUEST.ueNetworkCapability.uiaCap;

        v_Length := v_Length + 1;
      } else {

        v_SecurityCapability.uiaCap := omit;
      }

    } else {

      v_SecurityCapability.ueaCap := omit;

      v_SecurityCapability.uiaCap := omit; // @sic R5s100184 sic@

    }

    if (ispresent(p_ATTACH_REQUEST.msNetworkCapability)) {

      if (p_ATTACH_REQUEST.msNetworkCapability.gea1 == '1'B) {

        v_GEA := v_GEA or4b '1000000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea2 == '1'B) {

        v_GEA := v_GEA or4b '100000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea3 == '1'B) {

        v_GEA := v_GEA or4b '10000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea4 == '1'B) {

        v_GEA := v_GEA or4b '1000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea5 == '1'B) {

        v_GEA := v_GEA or4b '100'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea6 == '1'B) {

        v_GEA := v_GEA or4b '10'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea7 == '1'B) {

        v_GEA := v_GEA or4b '1'B;

      }

      v_Length := v_Length + 1;

      v_SecurityCapability.geaCap := bit2oct (v_GEA); // Now store the completed GEA octet

    } else {// MS Network Capability not present @sic R5s100184 sic@

      v_SecurityCapability.geaCap := omit;

    }

    v_SecurityCapability.iel := int2oct (v_Length, 1);  // Now store the length

    return v_SecurityCapability;

  }


After:

	  function f_EUTRA_FillNAS_SecurityCapability (ATTACH_REQUEST p_ATTACH_REQUEST) return UE_SecurityCapability

  {

    var UE_SecurityCapability v_SecurityCapability;

    var B8_Type v_GEA := '00000000'B;

    var integer v_Length := 2;

    v_SecurityCapability.eeaCap := p_ATTACH_REQUEST.ueNetworkCapability.eeaCap;

    v_SecurityCapability.eiaCap := p_ATTACH_REQUEST.ueNetworkCapability.eiaCap;

    if (ispresent(p_ATTACH_REQUEST.ueNetworkCapability.ueaCap)) {

      v_SecurityCapability.ueaCap := p_ATTACH_REQUEST.ueNetworkCapability.ueaCap;

      //v_Length := v_Length + 1;

      //24.301 9.9.3.36: If octet 5 is included, then also octet 6 shall be included

      v_Length := v_Length + 2;
      if (ispresent(p_ATTACH_REQUEST.ueNetworkCapability.uiaCap)) {

        v_SecurityCapability.uiaCap := p_ATTACH_REQUEST.ueNetworkCapability.uiaCap;

        //v_Length := v_Length + 1;
      } else {

        //v_SecurityCapability.uiaCap := omit;

        v_SecurityCapability.uiaCap := '00'O;
      }

    } else {

      v_SecurityCapability.ueaCap := omit;

      v_SecurityCapability.uiaCap := omit; // @sic R5s100184 sic@

    }

    if (ispresent(p_ATTACH_REQUEST.msNetworkCapability)) {

      if (p_ATTACH_REQUEST.msNetworkCapability.gea1 == '1'B) {

        v_GEA := v_GEA or4b '1000000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea2 == '1'B) {

        v_GEA := v_GEA or4b '100000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea3 == '1'B) {

        v_GEA := v_GEA or4b '10000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea4 == '1'B) {

        v_GEA := v_GEA or4b '1000'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea5 == '1'B) {

        v_GEA := v_GEA or4b '100'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea6 == '1'B) {

        v_GEA := v_GEA or4b '10'B;

      }

      if (p_ATTACH_REQUEST.msNetworkCapability.gea7 == '1'B) {

        v_GEA := v_GEA or4b '1'B;

      }

      v_Length := v_Length + 1;

      v_SecurityCapability.geaCap := bit2oct (v_GEA); // Now store the completed GEA octet

      if(v_Length == 3)

      {

      
//UE did not report UxA Capabilities, so we set them to 0

      
v_Length := 5;

      
v_SecurityCapability.ueaCap := '00'O;

      
v_SecurityCapability.uiaCap := '00'O;

      }
    } else {// MS Network Capability not present @sic R5s100184 sic@

      v_SecurityCapability.geaCap := omit;

    }

    v_SecurityCapability.iel := int2oct (v_Length, 1);  // Now store the length

    return v_SecurityCapability;

  } 


3.2 Change 2  /  UL grant handling cleanup at HO for 13.2.1
	Function name
	f_TC_13_2_1_EUTRA

	Reason for change
	1. UL grant in the target cell should be configured after HO, since the Cell

Config with the HO preparation contains a one time grant for C-RNTI based HO

using common preamble, so the SR based UL grant handling should be configured

afterwards, see HO test cases: 8.2.4.x

2. UL grant in the source cell should not be stopped in the middle of the HO

procedure, since there is a race condition between the RRC Connection

Reconfiguration Complete and the Cell Config Confirm. Stopping the grant in the

source cell is not necessary at all, see again the HO TCs, like 8.2.4.2



	Summary of change
	The code is corrected for the module MultiLayer_Procedures.ttcn in function

f_TC_13_2_1_EUTRA



	TTCN module
	MultiLayer_Procedures.ttcn

	MCC-STF160
	


Before:

	    /* Step 6: The UE transmits a MeasurementReport message on Cell 1 to report event A3 with the measured RSRP and RSRQ value for Cell 2. */

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1,

                                      cr_MeasurementReport_13_2_1_Step6 ( v_PhysCellId_Cell2 ) ) );

    //Activate the default to receive periodical reports for Cell2.

    v_DefaultRef := activate( a_EUTRA_AddDefault_PeriodicalMeasurmentReport( eutra_Cell1,

                                                                             cr_MeasurementReport_13_2_1_Step6 ( v_PhysCellId_Cell2 ) ) );

    //Start UL grant transmission in Cell 2

    f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now ); //@sic R5s100288 sic@
    //PDCP count shall be kept in PDCP re-establishment for AM DRB

    f_EUTRA_SS_PdcpCount_Handover ( eutra_Cell1, //@sic R5s100288 sic@

                                    eutra_Cell2,

                                    { tsc_DRB1 } );

    /* Step 7: The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform intra frequency handover to Cell 2. */

    f_EUTRA_508RRC_IntraLTE_HO ( eutra_Cell1, eutra_Cell2 ); //@sic R5s100288 sic@

    f_EUTRA_StopULGrantTransmission ( eutra_Cell1, cs_TimingInfo_Now ); //@sic R5s100288 sic@
    /* Step 8: The UE transmit an RRCConnectionReconfigurationComplete message on Cell 2 to confirm the successful completion of the intra frequency handover. */

    SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell2,

                                       cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    //Deactivate the default to stop receiving periodical Measurement Report

    deactivate( v_DefaultRef );

    /* Step 9: The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context on Cell 2. */

    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU( eutra_Cell2,

                                            tsc_RbId_DRB1,

                                            cs_TimingInfo_Now,

                                            { crs_IPv4_AnyIcmpEchoReply_2 ( px_IPv4_Address,

                                                                            px_IPv4_Address )}));




After:

	    /* Step 6: The UE transmits a MeasurementReport message on Cell 1 to report event A3 with the measured RSRP and RSRQ value for Cell 2. */

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1,

                                      cr_MeasurementReport_13_2_1_Step6 ( v_PhysCellId_Cell2 ) ) );

    //Activate the default to receive periodical reports for Cell2.

    v_DefaultRef := activate( a_EUTRA_AddDefault_PeriodicalMeasurmentReport( eutra_Cell1,

                                                                             cr_MeasurementReport_13_2_1_Step6 ( v_PhysCellId_Cell2 ) ) );

    //PDCP count shall be kept in PDCP re-establishment for AM DRB

    f_EUTRA_SS_PdcpCount_Handover ( eutra_Cell1, //@sic R5s100288 sic@

                                    eutra_Cell2,

                                    { tsc_DRB1 } );

    /* Step 7: The SS transmits an RRCConnectionReconfiguration message on Cell 1 to order the UE to perform intra frequency handover to Cell 2. */

    f_EUTRA_508RRC_IntraLTE_HO ( eutra_Cell1, eutra_Cell2 ); //@sic R5s100288 sic@

    /* Step 8: The UE transmit an RRCConnectionReconfigurationComplete message on Cell 2 to confirm the successful completion of the intra frequency handover. */

    SRB.receive( car_SRB1_RrcPdu_IND ( eutra_Cell2,

                                       cr_508_RRCConnectionReconfigurationComplete ( tsc_RRC_TI_Def ) ) );

    //Deactivate the default to stop receiving periodical Measurement Report

    deactivate( v_DefaultRef );

    // Start UL grant transmission and maintenance of PUCCH synchronized in target cell

    f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now );
    /* Step 9: The SS transmits one IP packet to the UE on the DRB associated with the default EPS bearer context on Cell 2. */

    DRB.send ( cas_DRB_COMMON_REQ_PDCP_SDU( eutra_Cell2,

                                            tsc_RbId_DRB1,

                                            cs_TimingInfo_Now,

                                            { crs_IPv4_AnyIcmpEchoReply_2 ( px_IPv4_Address,

                                                                            px_IPv4_Address )}));




3.3 Change 3  /  Redundant reset in Test 8.2.4.2 
	Function name
	f_TC_8_2_4_2_EUTRA

	Reason for change
	There is a redundant SRB/DRB reset in TC 8.2.4.2 step 6 since the common function

f_EUTRA_508RRC_IntraLTE_HO already performs this reset.



	Summary of change
	The function f_EUTRA_SS_SRBs_DRB1_ResetNow (eutra_Cell1) is removed from step 6 in function f_TC_8_2_4_2_EUTRA.

	TTCN module
	RRC_Handover.ttcn

	MCC-STF160
	


Before:

	    // Step 6: Receive RRCConnectionReconfigurationComplete

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( v_RRC_TI ) ) );

    //SS automatically checks that UE transmit an RRCConnectionReconfigurationComplete using the security key derived from the currently active KeNB

    //* @verdict pass RRCConnectionReconfigurationComplete received on Cell 2

    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.2.4.2 Step 6");

    // Deactivate the default to stop receiving measurement reports for cell 2 on cell 1

    deactivate( v_DefaultRef );

    // Reset MAC and re-establish RLC of cell 1

    f_EUTRA_SS_SRBs_DRB1_ResetNow ( eutra_Cell1 );
    // Start UL grant transmission and maintenance of PUCCH synchronized in target cell

    f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now );

    // Step 7: Check RRC Connected

    f_EUTRA_508Check_ConnectedState ( eutra_Cell2 );


After:

	    // Step 6: Receive RRCConnectionReconfigurationComplete

    SRB.receive ( car_SRB1_RrcPdu_IND ( eutra_Cell2, cr_508_RRCConnectionReconfigurationComplete ( v_RRC_TI ) ) );

    //SS automatically checks that UE transmit an RRCConnectionReconfigurationComplete using the security key derived from the currently active KeNB

    //* @verdict pass RRCConnectionReconfigurationComplete received on Cell 2

    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Test Case 8.2.4.2 Step 6");

    // Deactivate the default to stop receiving measurement reports for cell 2 on cell 1

    deactivate( v_DefaultRef );

    // Start UL grant transmission and maintenance of PUCCH synchronized in target cell

    f_EUTRA_StartDefULGrantTransmission ( eutra_Cell2, cs_TimingInfo_Now );

    // Step 7: Check RRC Connected

    f_EUTRA_508Check_ConnectedState ( eutra_Cell2 );


3.4 Change 4  /  Wrong RachProcedureReconfig in Postamble of 7.1.2.1
	Function name
	f_TC_7_1_2_1_EUTRA

	Reason for change
	A C-RNTI based RACH Procedure Reconfiguration is normally used to prepare / configure a HO procedure. But in case of the Postamble of 7.1.2.1, it does not make sense as there is no RACH Procedure Reconfiguration needed in the postamble at all (see 8.2.4.x Handover TCs, too).
Since there won't be any RACH Procedure in the test case afterwards, only the

Postamble (in connected mode) will be called to do the cleanup. So this step is

not only redundant but also wrong. 



	Summary of change
	The RachProcedureReconfig is removed from the Postamble of function f_TC_7_1_2_1_EUTRA. in TTCN module MAC_712.ttcn. 

	TTCN module
	MAC_712.ttcn

	MCC-STF160
	


Before:

	    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    f_EUTRA_SS_ConfigRachProcedure (eutra_Cell2, omit, cs_RachProcedureConfig_CRNTI_Def(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell2),

                                                       f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell2)) );// @sic R5-103846 sic@

    // Preconfigures SS for RACH response for default RAR and contention resolution
    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble (eutra_Cell2, E2_CONNECTED );//@sic R5-103875 sic@

    //Remove the cell1, non camping

    //@sic R5-103847 taken care in postamble now sic@

    


After:

	    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    //Switch off UE and expect a DETACH REQUEST message

    f_EUTRA_Postamble (eutra_Cell2, E2_CONNECTED );//@sic R5-103875 sic@

    //Remove the cell1, non camping

    //@sic R5-103847 taken care in postamble now sic@

    


3.5 Change 5  /  Postamble cleanup for TC 7.1.4.16
	Function name
	f_TC_7_1_4_16_EUTRA

	Reason for change
	1. The Default UL grant handling is not set back in the Postamble of 7.1.4.16 and thus messes the log up with unnecessary UL padding Data. 
2. Further, in the Postamble of 7.1.4.16 the PHR reporting is switched off in a wrong order that causes a race condition between the RRC Connection Reconfiguration Complete and the PHR Reports.


	Summary of change
	1. Function f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now) is added to the beginning of the Postamble section.
2. Function f_ SS_ConfigPowerHeadRoomIndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Disable) is moved to the beginning of the Postamble section.



	TTCN module
	MAC_714.ttcn

	MCC-STF160
	


Before:

	    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    // disable PHR reporting

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_RadioResConfig(tsc_RRC_TI_Def,

                                                                                cs_RadioResourceConfigDedicated_MAC({ explicitValue := cs_MAC_MainConfiguration_PHR_Disable},

                                                                                                                    omit,

                                                                                                                    omit))));

    t_Watchdog.start;

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;

    // to configure SS to stop report reception of  Power head room report

    f_SS_ConfigPowerHeadRoomIndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);
    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED );//@sic R5-103875 sic@

  } // end of 7.1.4.16


After:

	    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    // to configure SS to stop report reception of  Power head room report

    f_SS_ConfigPowerHeadRoomIndMode (eutra_Cell1, tsc_L1Mac_IndicationMode_Disable);

    // configure SS for default grant allocation

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);
    // disable PHR reporting

    SRB.send(cas_SRB1_RrcPdu_REQ(eutra_Cell1,

                                 cs_TimingInfo_Now,

                                 cs_RRCConnectionReconfiguration_RadioResConfig(tsc_RRC_TI_Def,

                                                                                cs_RadioResourceConfigDedicated_MAC({ explicitValue := cs_MAC_MainConfiguration_PHR_Disable},

                                                                                                                    omit,

                                                                                                                    omit))));

    t_Watchdog.start;

    SRB.receive(car_SRB1_RrcPdu_IND(eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "");

    t_Watchdog.stop;

    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED );//@sic R5-103875 sic@

  } // end of 7.1.4.16


3.6 Change 6  /  Postamble cleanup for TC 7.1.4.15
	Function name
	f_TC_7_1_4_15_EUTRA

	Reason for change
	1. The Default UL grant handling is not set back in the Postamble of 7.1.4.15 and thus messes the log up with unnecessary UL padding Data. 


	Summary of change
	1. Function f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now) is added to the beginning of the Postamble section.



	TTCN module
	MAC_714.ttcn

	MCC-STF160
	


Before:

	    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    // to configure SS to stop report reception of  Power head room report

    f_SS_ConfigPowerHeadRoomIndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable);

    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED );//@sic R5-103875 sic@

  } // end of

  


After:

	     ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    // to configure SS to stop report reception of  Power head room report

    f_SS_ConfigPowerHeadRoomIndMode (eutra_Cell1,tsc_L1Mac_IndicationMode_Disable);

    // configure SS for default grant allocation

    f_EUTRA_StartDefULGrantTransmission(eutra_Cell1, cs_TimingInfo_Now);
    // Release RRC Connection and Switch off

    f_EUTRA_Postamble(eutra_Cell1, E2_CONNECTED );//@sic R5-103875 sic@

  } // end of

  


3.7 Change 7  /  Correction of UL Cell Config for Band 20
	Function name
	f_EUTRA_InitFrequencyLowMidHigh 

	Reason for change
	Incorrect UL frequency settings for Band 20.

	Summary of change
	Corrected UL frequency settings for Band 20 acc. to 36.508

	TTCN module
	EUTRA_CellInfoFrequency.ttcn

	MCC-STF160
	


Before:

	      case (20) {   // Band 20  @sic R5-100846 sic@

        select (p_DL_ChBandwidth){

          case (n25){  // ChBandwidth = 5

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6175;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6425;

          }

          case (n50){  // ChBandwidth = 10

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6200;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6400;

          }

          case (n75){  // ChBandwidth = 15

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6225;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6375;

          }

          case (n100){  // ChBandwidth = 20

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6250;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6350;

          }

          case else{

            FatalError (__FILE__, __LINE__, "invalid ChannelBandwidth");

          }

        }

        select (p_UL_ChBandwidth) {

          case (n25){  // ChBandwidth = 5

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 22775;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 22875;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 22975;

          }

          case (n50){  // ChBandwidth = 10

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 22800;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 22875;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 22950;

          }

          case (n75){  // ChBandwidth = 15

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 22825;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 22875;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 22925;

          }

          case (n100){  // ChBandwidth = 20

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 22850;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 22875;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 22900;

          }

          case else{

            FatalError (__FILE__, __LINE__, "invalid ChannelBandwidth");

          }

        }

      }


After:

	      case (20) {   // Band 20  @sic R5-100846 sic@

        select (p_DL_ChBandwidth){

          case (n25){  // ChBandwidth = 5

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6175;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6425;

          }

          case (n50){  // ChBandwidth = 10

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6200;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6400;

          }

          case (n75){  // ChBandwidth = 15

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6225;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6375;

          }

          case (n100){  // ChBandwidth = 20

            v_FreqLMH.FrequencyLow.dl_CarrierFreq := 6250;

            v_FreqLMH.FrequencyMid.dl_CarrierFreq := 6300;

            v_FreqLMH.FrequencyHigh.dl_CarrierFreq := 6350;

          }

          case else{

            FatalError (__FILE__, __LINE__, "invalid ChannelBandwidth");

          }

        }

        select (p_UL_ChBandwidth) {
          case (n25){  // ChBandwidth = 5 

            v_FreqLMH.FrequencyLow.ul_CarrierFreq :=  24175;  // See 36.508 clause 4.3.1.1.20

            v_FreqLMH.FrequencyMid.ul_CarrierFreq :=  24300;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 24425;
          }

          case (n50){  // ChBandwidth = 10

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 24200;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 24300;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 24400;
          }

          case (n75){  // ChBandwidth = 15

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 24225;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 24300;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 24375;
          }

          case (n100){  // ChBandwidth = 20

            v_FreqLMH.FrequencyLow.ul_CarrierFreq := 24250;

            v_FreqLMH.FrequencyMid.ul_CarrierFreq := 24300;

            v_FreqLMH.FrequencyHigh.ul_CarrierFreq := 24350;

          }

          case else{

            FatalError (__FILE__, __LINE__, "invalid ChannelBandwidth");

          }

        }

      }


3.8 Change 8  /  TFTs / Packet Filter IDs

	Function name
	f_Get_508_PacketFilter

	Reason for change
	Packet filter ID for reference packet filter #2 has same id=0 as packet filter #1. Packet filter IDs irrespective of DL and UL shall be different. A prose CR to 36.508 is also needed and expected in next RAN5 #48 meeting, August 2010.

Alternative proposal compared to R5s100287.

	Summary of change
	Packet filter id set to 2, which results in iei=’22’O

	TTCN module
	NAS_AuxiliaryDefsAndFunctions.ttcn

	MCC-STF160
	


Before:

	  function f_Get_508_PacketFilter(integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId) return template (value) PacketFilter

  {

    var integer v_FilterCount := 0;

    var template (value) PacketFilter v_Filter;

    if (pc_IPv4){

      v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress;

      v_FilterCount := v_FilterCount + 9;

    }

    else { // IPv6

      v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress;

      v_FilterCount := v_FilterCount + 33;

    }

    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.iei         := '10'O;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2;

      }

      case (2) {

        // Update to values in Reference packet filter #2

        v_Filter.iei         :=  '20'O;
        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2

          }

      case (3){

        // Update to values in Reference packet filter #3

        v_Filter.iei         :=  '31'O; // Bidirectional filter, ID=1

        v_Filter.precedence  := '00001111'B;

      }

    }

    v_Filter.iel := int2oct(v_FilterCount, 1);

    return v_Filter;

  }            

                 


After:

	  function f_Get_508_PacketFilter(integer p_PacketFilterNumber,

                                  HalfOctet_Type p_EPS_BearerId) return template (value) PacketFilter

  {

    var integer v_FilterCount := 0;

    var template (value) PacketFilter v_Filter;

    if (pc_IPv4){

      v_Filter.contents[0] := cs_PktFilterCompIPv4RemoteAddress;

      v_FilterCount := v_FilterCount + 9;

    }

    else { // IPv6

      v_Filter.contents[0] := cs_PktFilterCompIPv6RemoteAddress;

      v_FilterCount := v_FilterCount + 33;

    }

    select (p_PacketFilterNumber){

      case (1) {

        v_Filter.iei         := '10'O;

        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort31160(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2;

      }

      case (2) {

        // Update to values in Reference packet filter #2

        v_Filter.iei         :=  '22'O;
        v_Filter.precedence  := int2bit((hex2int(p_EPS_BearerId)-6), 8);

        v_Filter.contents[1] := cs_PktFilterCompSingleRemotePort61000(p_EPS_BearerId);

        v_FilterCount := v_FilterCount + 3;

        v_Filter.contents[2] := cs_PktFilterCompProtocolIdNextHeader;

        v_FilterCount := v_FilterCount + 2

          }

      case (3){

        // Update to values in Reference packet filter #3

        v_Filter.iei         :=  '31'O; // Bidirectional filter, ID=1

        v_Filter.precedence  := '00001111'B;

      }

    }

    v_Filter.iel := int2oct(v_FilterCount, 1);

    return v_Filter;

  }


3.9 Change 9  /  Correction to Attach Procedure in EUTRA NAS Common module
	Function name
	f_AttachProcedure

	Reason for change
	In test case 9.2.1.1.14, the function is used at step 19 after an attach reject procedure. In this case the NAS security context is removed, i.e. the UE will send its first Attach Request not integrity protected.
Alternative proposal compared to R5s100358.

	Summary of change
	The check of integrity protection is removed

	TTCN module
	EUTRA_NASCommonFunctions.ttcn

	MCC-STF160
	


Before:

	  function f_AttachProcedure ( CellId_Type p_CellId,

                               charstring p_VerdictComment,

                               EUTRA_ATTACH_Type p_ForcedAttach := NORMAL ) runs on EUTRA_PTC

  {

    /* UE Registration acc. to 36.508 cl. 4.5.2 */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_AttDetValue_Type v_AttachType :=f_GetEAttachType();

    var B3_Type v_PDNType := f_GetPdnType();

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    // Receive RRC Connection Setup Complete with

    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                v_RRC_TI,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_IndicationWithPiggybacking(tsc_SHT_IntegrityProtected,

                                                                                  cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType)));

    if (p_VerdictComment != "No Verdict") {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_VerdictComment);

    }

    f_EUTRA_IdleUpdated_Step5_17 (p_CellId, STATE2_IDLEUPDATE, p_ForcedAttach, v_NAS_Ind);

  } // f_AttachProcedure


After:

	  function f_AttachProcedure ( CellId_Type p_CellId,

                               charstring p_VerdictComment,

                               EUTRA_ATTACH_Type p_ForcedAttach := NORMAL ) runs on EUTRA_PTC

  {

    /* UE Registration acc. to 36.508 cl. 4.5.2 */

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_AttDetValue_Type v_AttachType :=f_GetEAttachType();

    var B3_Type v_PDNType := f_GetPdnType();

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    // Receive RRC Connection Setup Complete with

    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                v_RRC_TI,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_IndicationWithPiggybacking(?,

                                                                                  cr_508_ATTACH_REQUEST(v_AttachType, v_AdditionalUpdateType),

                                                                                  cr_508_PDN_CONNECTIVITY_REQUEST(tsc_PdnRequest_InitialAttach, v_PDNType)));

    if (p_VerdictComment != "No Verdict") {

      f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_VerdictComment);

    }

    f_EUTRA_IdleUpdated_Step5_17 (p_CellId, STATE2_IDLEUPDATE, p_ForcedAttach, v_NAS_Ind);

  } // f_AttachProcedure

